
How to Use a Permanent SAS Data Set
(commands=useperm.sas)

Introduction:

This chapter discusses using permanent SAS data sets from different releases of SAS on Windows. In general, SAS data sets are downwardly compatible across releases (e.g. SAS 9, SAS 8, SAS 7, SAS 6); a later release of SAS can generally read data sets from an earlier release by specifying the correct engine to read the data set. 
The SAS data sets discussed in this handout are contained in two Zipped Files: SASDATA1.ZIP (Version 6 SAS data sets) and SASDATA2.ZIP (Version 8/9 SAS data sets). These two zipped files can be found on my web page: http://www.umich.edu/~kwelch. Specific information on the contents of each of these data sets can be obtained in the handout “Data Descriptions.doc”, also on my web page.

SASDATA1.ZIP contains the following version 6 SAS data sets:

· FITNESS.SD2

· GPA.SD2

· MARCH.SD2

· SURVEY.SD2

SASDATA1.ZIP also contains a version 6 formats catalog, FORMATS.SC2, which will not be discussed in this handout.

SASDATA2.ZIP contains the following SAS version 8/9 data sets:

· autism_demo.sas7bdat

· autism_socialization.sas7bdat

· bank.sas7bdat

· baseball.sas7bdat

· business.sas7bdat

· employee.sas7bdat

· iris.sas7bdat

· tecumseh.sas7bdat

· wave1.sas7bdat

· wave2.sas7bdat

· wave3.sas7bdat

· ship.sd7 (version 7/8 short file extension)

Plus, a SAS transport file:

· owen.xpt.

The ship.sd7 data set has what is called a “short file extension”. The name of this file will need to be changed to ship.sas7bdat before it can be read using SAS release 9.

Some definitions:

Library:

A library is a location on your computer (i.e.,. a folder or directory) where SAS data sets are stored. Because a library refers to the entire folder (not to an individual data set), one library can have several data sets stored in it, and it is possible for them to be of mixed types. However, a particular engine assigned to a given folder will only "SEE" files of one type. It is good practice to keep SAS data sets of different types in different folders.

Default Library: Work, the temporary library, is the default library that SAS assumes if no libname is specified for a data set. .  
Engine:

An engine tells SAS the type of files it is to read. Some engines that you can use are: 

· V9 the default engine for SAS release 9 (.sas7bdat) data sets (the V8 engine works, too)

· V8 the default engine for SAS release 8 (.sas7bdat) data sets

· V6 the engine used to read/write SAS release 6.08 through 6.12 (.sd2) data sets

· V604 to read  (but not write) data sets created using PC SAS release 6.04 (.ssd) data sets

Default engine: If you do not assign an engine to a library, the default engine will be the engine corresponding to the release of the data sets found in the folder. If there are no data sets in a folder, the default engine will be the the current version of SAS; i.e., if you are running SAS 9.1 or 9.2, SAS will automatically use the V9 engine, etc.
If there are data sets of mixed types within a folder, SAS will assign the engine corresponding to the highest version compatible with any of the data sets in the folder. So if there are both V6 and V9 data sets in a folder, SAS will assign the V9 engine to read data from that folder. To read V6 data sets from a folder containing both V6 and V9 data sets, the V6 engine must be used explicitly. To avoid confusion, we highly recommend including only one type of data sets within any given folder. 

Step-By-Step Instructions for Using a Permanent SAS Data Set:

1. Determine the File type:  The first step in using a SAS data set is to determine what type of file it is (i.e. the operating system from which it originated and the SAS release or engine used to create it). The table below shows the file extensions used by SAS to distinguish between data sets created on different operating systems using different releases of SAS.

Operating System
SAS Release
            Extension
Example

Windows                    V7.0  to  V9.0
              .sas7bdat*       business.sas7bdat

Windows                    V7.0   to V8.0
              .sd7**             ship.sd7

Windows

V6.08 to V6.12 
.sd2     

fitness.sd2

Unix


V6.06 to V6.12
.ssd01      
mydata.ssd01

Macintosh

V6.10 to V6.12
.ssd01  
mydata.ssd01  

DOS


V6.04
(PC SAS)
.ssd

mydata.ssd

  * .sas7bdat  is the default data set extension for SAS Windows release 7, 8, and 9..

**  .sd7 extension files cannnot be read by SAS Windows release 9. If you have SAS data sets that end in the .sd7 file extension, rename them to .sas7bdat before trying to use them in SAS.
(NB: If you cannot see the file extensions of the SAS data sets on your computer: Go to Windows Explorer or My Computer. Select Tools…Folder Options…View. Make sure that “Hide file extensions for known file types” is NOT selected.)
2. Assign a Library Name and Engine, using either a Libname Statement, or the Assign New Libraries Icon:  
Assigning a libref using a Libname statement:

The libname statement assigns an alias (called a libref) to a directory that you specify. The directory must already exist The libref name that you assign must be 8 characters or less to be valid in SAS. You can assign any number of libname statements in a given SAS session. Two examples of  libname statements are shown below:

libname sasdata2 V9 "c:\sasdata2";

libname sasdata1 V6 "c:\sasdata1";

In the first libname statement, the libref, SASDATA2, and the V9 engine are assigned to the folder: c:\sasdata2. This will allow SAS to read any SAS release 8/9 (.sas7bdat) files included in this folder. 

In the second libname statement, the libref, SASDATA1, and the V6 engine are assigned to the folder: c:\sasdata1. This will allow SAS to read any SAS release 6 (.sd2) files included in this folder.

NB: Be sure to highlight and submit the libname statement for it to take effect. The libname statement will remain in effect for your entire SAS session. If you restart SAS, you will need to resubmit the libname statement. The libname does not have to be the same as the folder where you have stored your dataset, although it is helpful to remember it that way.
Assigning a library using the New Library icon:

Click on the New Library icon, and in the Name: Dialog box, specify the name of the library you wish to assign. Choose the appropriate Engine from the drop-down menu. Then browse to the location of the folder you wish to assign. To make sure a library will still be assigned in a later session, select “Enable at startup”. This option will work on your personal computer, but will not take effect at the Public Computing Sites.

3. Specify a two-level name for the permanent SAS data sets you wish to use in any Proc Steps or Data Steps, as shown in the examples below.  The two-level name is of the form libname.datasetname. Note that there are no spaces between the libname and the dataset name. 
Examples:

Suppose you have unzipped the data sets in SASDATA2.ZIP to the folder c:\sasdata2. Here is how you can submit a libname statement so you can utilize the datasets in this folder.
libname sasdata2 V9 "c:\sasdata2";

SAS will not produce any output in the output window as a result of submitting these commands, but you will see the following note in the SAS Log. 

libname sasdata2 V9 "c:\sasdata2";

NOTE: Libref SASDATA2 was successfully assigned as follows:

      Engine:        V9

      Physical Name: c:\sasdata2

Once the libname statement has been submitted (no run statement is necessary), you will be able to use any of the SAS release 9 data sets in the c:\sasdata2 folder. You will need to specify the data set to use with the data= option for each procedure. The libname statement will be in effect for the entire SAS session, and so it only needs to be submitted once.

title "Business data set";

proc means data=sasdata2.business;

run;

title "Iris data set";

proc means data=sasdata2.iris;

run;

If you wish to use the version 6 (.sd2) data sets that have been unzipped into the c:\sasdata1 folder, you will need to submit another libname statement, and then use the two-level name of any data set you wish to use.

libname sasdata1 V6 "c:\sasdata1";

title "GPA data set";

proc means data=sasdata1.gpa;

run;

title "Fitness data set";

proc means data=sasdata1.fitness;

run;

To get the contents of all version 9 SAS data sets in the sasdata2 library.:
proc contents data=sasdata2._all_ ;

run;

Make sure there are no blanks in the sasdata._all_  portion of your command. You can use similar commands to view the contents of the sasdata1 library.

Assigning a Default Data Set:
A default data set can be assigned with an options _last_= statement after a libname statement. (Be sure you have no blanks in _last_.)  This allows you to utilize the same data set without having to specify it for each procedure. In the example below, sasdata1.fitness will be used for all procedures.

options _last_=sasdata2.baseball; 

title "SASDATA2.BASEBALL Data Set";
proc means;

run;

proc freq;

  tables team;

run;

proc reg;

  model salary = cr_home;

run; 

quit;

The default data set will be in effect until a new one is specified with another options statement, or until another new data set is created.

Note on Using Permanent Data Sets in SAS Release 8 and 9:

You can specify permanent SAS data sets to use by giving the complete path and file name in quotes, starting with SAS release 8. This avoids the libname statement, but does not allow a default data set to be specified. Data set options (e.g., obs= ) can still be specified in parentheses after the quoted file name. 

title "SASDATA2.IRIS Data Set";

proc freq data="c:\sasdata2\iris.sas7bdat"; 

  tables species;

run; 
proc print data="c:\sasdata2\iris.sas7bdat"(obs=10);

run;

How to Open a Permanent SAS Data set in SAS/INSIGHT:

Activate SAS/INSIGHT. In the dialog box that appears choose the library that you wish to use (e.g. SASDATA2), then double-click on the data set that you wish to open (e.g. BUSINESS). To open another data set, simply go to File…Open…and choose the data set to open.

How to create a temporary SAS data set from a permanent one:

Many SAS users simply create a temporary SAS data set to use in a given session.  This temporary data set becomes the default automatically.

data business;

  set sasdata2.business; 

run;

title "Business data set";

proc means data=business; 

run;

This method has the advantage of allowing you to work with a temporary SAS data set, which is often simpler than working with a permanent one. But it can be cumbersome if you have a large data set, because it creates a whole new copy of the data in the WORK library. 

How to de-assign a library:

Use the libname statement with the option clear to de-assign a library. The library assignment will be cleared, but the data sets in the library will not be affected. Do not specify an engine here.
libname sasdata1 clear;

Automatically assign libnames using the Autoexec.sas file:

The library or libraries that you wish to use must be re-assigned for each session, if you assign them using a libname statement. To have SAS remember your libraries from one run to another, you can create a file called autoexec.sas, and place the libname statements in it, as shown below.  Each time SAS starts up, it will read the autoexec.sas file, and assign the appropriate libraries. 

libname sasdata1 V6 "c:\sasdata1";

libname sasdata2 v9 "c:\sasdata2";

If you place the autoexec.sas file in the directory from which SAS is running, SAS will read it and execute the commands it contains each time it starts up. However, if you save the autoexec.sas file in another location, you can specify it as an option in the SAS shortcut. An example SAS shortcut is shown below, followed by the notes in the SAS Log.

"C:\Program Files\SAS\SAS 9.1\sas.exe" -CONFIG "C:\Program Files\SAS\SAS 9.1\nls\en\SASV9.CFG" –AUTOEXEC “c:\temp\autoexec.sas”

NOTE: AUTOEXEC processing beginning; file is c:\temp\autoexec.sas.

NOTE: Libref SASDATA1 was successfully assigned as follows:

      Engine:        V6

      Physical Name: c:\temp\sasdata1

NOTE: Libref SASDATA2 was successfully assigned as follows:

      Engine:        V9

      Physical Name: c:\temp\sasdata2

NOTE: AUTOEXEC processing completed
 How to Create a Permanent SAS Data Set

(commands=saveperm.sas)

Introduction:

A permanent SAS data set is saved to a location where it can be retrieved and used later, without having to recreate it each time you restart SAS. In addition, transformations, recodes and other data manipulations are saved and do not need to be re-run every time the data set is used. Several people can share the same permanent data set over a network. 

There are two steps necessary to create a permanent SAS data set:

· Assign a library and engine.

· Create the data. Be sure to assign both a library (other than WORK) and data set name to make the data set permanent.

A library is a location on your computer (e.g. a folder or directory) where SAS data sets and other SAS files are stored. A library usually refers to the entire folder and not to individual data sets. One library can have several data sets stored in it. The libname statement is used to define a library. An engine tells SAS the type of files that it is to write. See Chapter 8 for more information about libraries and engines.

It is highly recommended that you store only one type of SAS data set in a given folder.

Create a Permanent SAS data set using a Data Step:

Suppose you wish to store your SAS data sets in the c:\temp\sasdata2 directory. First submit a libname statement from the program editor. The libname statement assigns a name (called a libref) to the directory that you specify. 

Note: the libname statement must point to a folder that already exists.  Be sure to create the folder if it does not already exist.

libname sasdata2 V9 "c:\sasdata2";

data sasdata2.pulse;

   infile "pulse.dat";

   input pulse1 pulse2 ran smokes sex height weight activity;

run;

Sasdata2.pulse, will contain all variables originally read into SAS using the input statement, plus any new variables that you create. It will now be the default dataset.
Create a Permanent SAS data set using the Import Wizard:
To make a permanent SAS data set using the Import Wizard, you must first submit a libname statement from the Program Editor Window (see discussion of the libname statement above). The data set can then be saved in the (pre-defined) library in the “Select library and member” window of the Import Wizard. For example, the libname statement below can be submitted from the Program Editor Window to define the sasdata2 library. 

libname sasdata2 V9 "c:\sasdata2";

From the pull-down menu in the Library box, choose SASDATA2 as the library. Then type the data set name, PULSE, in the Member dialog box:

The data set sasdata2.pulse will now be the default, because it was the most recent one created by SAS in this current session. It can be used without referring to its name in the current session.

proc means;

run;

proc freq;

   tables sex ran smokes;

run;

Or you can refer to the data set using its two-level name by specifying a data= option. 
proc means data=sasdata2.pulse;

run;

proc freq data=sasdata2.pulse;

   tables sex ran smokes; 

run;

Create a Permanent SAS data set using Proc Import:
You can also import an Excel file using Proc Import syntax. Type the two-level name as the value for the out= keyword, as shown below. (The syntax below was saved from importing the file PULSE.XLS using the Import Wizard.)

libname sasdata2 V9 "c:\sasdata2";
PROC IMPORT OUT= SASDATA2.PULSE 

            DATAFILE= "c:\temp\labdata\PULSE.XLS" 

            DBMS=EXCEL REPLACE;

     SHEET="pulse$"; 

     GETNAMES=YES;

     MIXED=NO;

     SCANTEXT=YES;

     USEDATE=YES;

     SCANTIME=YES;

RUN;

Create a Permanent SAS Data Set as Output from Another Procedure:
Many SAS procedures can create output data sets to be used later. For example, when running Proc Reg, an output data set can be created containing the predicted values and residuals from a fitted model. The commands below show how to create a permanent SAS data set, named sasdata2.resids, as output from Proc Reg. Note that the libname statement must be submitted first:

libname sasdata2 V9 "c:\temp\labdata";

proc reg data=sasdata2.pulse;

  model pulse2 = pulse1 ;

  output out = sasdata2.resids p=predict r=resid rstudent=rstudent;

run;

quit;

The following note is produced in the SAS Log:

180  proc reg data=sasdata2.pulse;

181    model pulse2 = pulse1 ran;

182    output out = sasdata2.resids p=predict r=resid rstudent=rstudent;

183  run;

183  quit;

NOTE: The data set SASDATA2.RESIDS has 92 observations and 15 variables.

NOTE: PROCEDURE REG used (Total process time):

      real time           0.06 seconds

      cpu time            0.06 seconds
The sasdata2.resids data set can now be used to check the distribution of the residuals, using Proc Univariate, as shown below:

proc univariate data=sasdata2.resids;

  var resid;

  histogram;

  qqplot / normal (mu=est sigma=est);

run;

Note that the data set, sasdata2.resids will now be the default data set, because it was the most recently created data set in the current session of SAS.
How to Use a Permanent SAS Data Set in Later Runs of SAS:
To use a permanent SAS data set in later runs of SAS, you must submit a libname statement, and refer to the data set by its two-level name:

libname sasdata2 V9 "c:\sasdata2";

proc means data=sasdata2.pulse;

run;

proc freq data=sasdata2.pulse;

   tables ran smokes;

run;
How to delete a permanent SAS data set:

There are 3 basic ways to do delete a permanent SAS data set. 1) Go to the SAS Explorer and delete the files by right-clicking a file name and choosing delete. 2) Go to the Windows Explorer and delete the SAS data sets. 3) Use Proc Datasets, as shown in the example below:

libname sasdata2 V9 "c:\sasdata2";
proc datasets library=sasdata2;

  delete pulse;

  delete resids;

run; 

quit;

This procedure does not produce any output, but generates the following information in the SAS Log:

217 proc datasets library=sasdata2;

                                           Directory

                                Libref         SASDATA2

                                Engine         V9

                                Physical Name  c:\temp\sasdata2

                                File Name      c:\temp\sasdata2

                               Member      File

                  #  Name      Type        Size  Last Modified

                  1  BASEBALL  DATA       82944  20Jun02:07:12:32

                  2  BUSINESS  DATA       17408  20Jun02:07:12:32

                  3  IRIS      DATA       13312  20Jun02:07:12:32

                  4  PULSE     DATA       17408  14Aug06:16:00:17

                  5  RESIDS    DATA       25600  14Aug06:16:00:25

                  6  TECUMSEH  DATA     1147904  02Jun05:00:00:04

218    delete pulse;

219    delete resids;

220  run;

NOTE: Deleting SASDATA2.PULSE (memtype=DATA).

NOTE: Deleting SASDATA2.RESIDS (memtype=DATA).

221  quit;

NOTE: Deleting SASDATA2.PULSE (memtype=DATA).

198  quit;

NOTE: PROCEDURE DATASETS used (Total process time):

      real time           0.11 seconds

      cpu time            0.01 seconds
These data sets will be permanently deleted from the folder. To get them back, you must explicitly re-create them.

You can also delete data sets interactively by going to the libraries window and right-clicking on the file, and then selecting delete. This will also permanently delete a data set.

Overview of Data Management Tasks

(command file=datamgt.sas)
Create the March data set:
To create the March data set, you can read it from the MARCH.DAT raw data file, using a data step, as shown below.

data march;

  infile "marflt.dat";

  input flight 1-3
           @4 date mmddyy6.



@10 time time5.



orig $ 15-17


dest $ 18-20


@21 miles comma5.



mail 26-29


freight 30-33


boarded 34-36


transfer 37-39


nonrev 40-42


deplane 43-45


capacity 46-48;

format date mmddyy10. time time5. miles comma5.; 

run;

Or, you can import the March.xls file from Excel, using the Import Wizard (commands not shown)..

Make a copy of a SAS data set:

You can use a set statement to make a copy of a data set. In the commands below, NEWMARCH is created by making an exact copy of MARCH (which we assume was created as a temporary data set in the current SAS session). 

data newmarch;

   set march;

run;

A set statement can also be used to make a copy of a permanent SAS data set:

libname sasdata2 "c:\sasdata2";

data sasdata2.business2;

     set sasdata2.business;

run;

Additional commands can be added to the data step to create new variables, or to modify the data set in other ways. 

data newmarch;

   set march;

   /*additional SAS statements*/

run;

NB: be sure all changes that you wish to make to your new data set are included before the run statement! After the run statement, the data set will be closed, and no additional variables can be added, or changes made to the data set. 
Create a subset of data:
You can easily create a subset of your data by using the set statement along with a subsetting if statement. The subsetting if statement acts as a gateway for allowing observations to be written to a data set. In the examples below, the data set named MARCH15 will contain information on flights only on March 15th, 1990, while the data set named LONDON will contain information on all flights to London, and the data set named LONGFLT will contain information on all flights of 1000 miles or more..

data march15;

   set march;

   if date = "15MAR1990"D;

run;

data london;

   set march;

   if dest="LON";

run;

data longflt;

   set march;

   if miles >=1000;

run;

NB: The subsetting if can be used at any place in your data step code. It will only take effect when the data set is written out. 

Another way to select cases to be included in a data set is to use an output statement. It is important to note that the output statement takes effect immediately (at the point in your code where it is included). Any commands that are added after the output statement will not affect the cases that were output earlier.

WRONG:

data london_latemarch;

   set march;

   if dest="LON" and date >="15MAR1990"D then output;

   totpassngrs = boarded + transfer + nonrev;

   pctfull = (totpassngrs/capacity)*100;

run;

The data set LONDON_LATEMARCH will contain flights to London on or after March 15th, the new variables TOTPASSNGRS and PCTFULL will be included in the new data set, but they will not have any values in them, because they were defined after the output statement. To correct this problem, make the output statement the last statement in the data step.
RIGHT:

data london_latemarch2;

   set march;
   totpassngrs = boarded + transfer + nonrev;

   pctfull = (totpassngrs/capacity)*100;

   if dest="LON" and date >="15MAR1990"D then output;

run;

Delete CASES from a data set:
A delete statement can be used to remove a case or cases from a data set. When the case is deleted, it is permanently removed from the data set. The delete statement takes effect immediately when it is specified, so deleted cases will not be available for any later programming statements.

data shortflt;

   set march;

   if miles >=1000 then delete;

   if date=. then delete;

run;
Similar to the output statement, the delete statement takes effect at the point in the data step where it is placed.

Keep or Drop VARIABLES:
You can control the variables that are included in a SAS data set by using keep and drop statements as part of the data step. The keep and drop statements may be given at any point in the data step, and only take effect at the time the data set is written. 

data march_passngrs;

   set march;

   keep date time orig dest miles boarded transfer nonrev deplane capacity;

run;
data march_passngrs2;

   set march;

   drop mail freight;

run;
Create new variables using transformations and recodes:

New variables can be created using transformations and recodes of variables in the data step using assignment statements with SAS functions, or if…then statements. Assignment statements (shown below) are used to create new variables based on the value of previously defined variables, expressions, or constants, and are of the form: 

newvar = expression;

Example:

data march_recode;

   set march;

   totpassngrs = boarded + transfer + nonrev;

   empty_seats = capacity – totpassngrs;

   totnonpass  = mail + freight;

   pctfull = (totpassngrs/capacity)*100;

   logpassngrs = log(totpassngrs);

   totpassngrs2 = sum(boarded,transfer,nonrev);

   int_pctfull = int(totpassngrs/capacity)*100;

   rnd_pctfull = round(pctfull,.1); 

   if pctfull = 100 then full_flight = 1;

   else full_flight = 0;

   if pctfull = . then full_flight = .;

   if pctfull not=. then do;

      if pctfull < 25 then full_cat = 1;

      if pctfull >=25 and pctfull <50 then full_cat=2;

      if pctfull >=50 and pctfull <75 then full_cat=3;

      if pctfull >=75 then full_cat=4;

   end;

   if dest = "CPH" or dest="FRA" or dest = "LON" 

     or dest = "PAR" or dest = "YYZ" then USA = 0;

   if dest in("DFW", "LAX", "ORD", "WAS") then USA = 1;

run;  

Check the new variables using Proc Means and Proc Freq:

title "Check New Variables";

proc means data=march_recode;run;

proc freq data=march_recode;

  tables full_flight full_cat dest USA;

run;

                                      Check New Variables     

                                      The MEANS Procedure

 Variable       Label             N           Mean        Std Dev        Minimum        Maximum

 ----------------------------------------------------------------------------------------------

 flight         Flight number   635    447.9086614    275.3102085    114.0000000    982.0000000

 date                           634       11031.98      8.9801263       11017.00       11047.00

 time                           635       47952.47       14707.08       24960.00       75960.00

 miles                          635        1615.25        1338.47    229.0000000        3857.00

 mail                           634    381.0031546     74.6288128    195.0000000    622.0000000

 freight                        634    333.9511041     98.1122248     21.0000000    631.0000000

 boarded                        633    132.3570300     43.4883098     13.0000000    241.0000000

 transfer                       635     14.4062992      5.3362008              0     29.0000000

 nonrev                         635      4.1133858      1.9243731              0      9.0000000

 deplane                        635    146.7842520     45.4289656     18.0000000    250.0000000

 capacity                       635    205.3795276     27.1585929    178.0000000    250.0000000

 totpassngrs                    633    150.8878357     43.0930520     31.0000000    250.0000000

 empty_seats                    633     54.5244866     34.9192529              0    151.0000000

 totnonpass                     633    715.1927330    124.8981261    341.0000000        1085.00

 pctfull                        633     73.0774908     17.7696598     17.2222222    100.0000000

 logpassngrs                    633      4.9681880      0.3292127      3.4339872      5.5214609

 totpassngrs2                   635    150.4598425     43.6959260      9.0000000    250.0000000

 int_pctfull                    633      8.3728278     27.7198922              0    100.0000000

 rnd_pctfull                    633     73.0764613     17.7693610     17.2000000    100.0000000

 full_flight                    633      0.0837283      0.2771989              0      1.0000000

 full_cat                       633      3.3791469      0.6785651      1.0000000      4.0000000

 USA                            632      0.6819620      0.4660832              0      1.0000000

 ----------------------------------------------------------------------------------------------
The two variables, TOTPASSNGRS (n=633) and TOTPASSNGRS2(n=635) have different numbers of cases, because they were created in different ways. TOTPASSNGRS is created using the mathematical operators (+), so the resulting variable is missing if the value of any of the variables in the expression is missing. TOTPASSNGRS2 returns the sum of the non-missing argument variables, so it produces the sum of any non-missing argument variables. 

Also note the syntax used to create the new variables, FULL_FLIGHT and FULL_CAT. When using “Else” with SAS, all other values, including missing will be included in the Else category. We get around this by setting cases with a missing value for PCTFULL to missing in the resulting variable. We use an If…then statement when creating FULL_CAT to be sure this new variable is only created if PCTFULL is not missing (if pctfull not=. then do;). If you use an if…statement, it must be followed by an end statement. 

The new variable USA is created from the character variable, DEST. The in operator is used to shorten the syntax for setting up the value of USA=1. 
Sort Cases:
Use Proc Sort to sort a data set. Once sorted, a data set remains sorted, and any later analyses can be done either for the entire data set, or for subgroups by including a by statement in a procedure. A separate analysis will be done for each of the "by" groups.  

proc sort data=march_recode;

  by USA; run;
title “Descriptive Statistics by US vs Non-US Destinations”;

proc means data=march_recode;

  by USA; run;
Sorting by more than one variable:

You can sort by several variables, as shown in the example below. Proc sort organizes the data so that the first variable represents the slowest changing index (i.e., cases will be sorted first by DATE, and then by levels of  DEST within DATE). 

proc sort data=march_recode;

     by date USA; 

run;
title “Descriptive Statistics by Date and Destination”;

proc means data=march_recode;

     by date USA; 

run;
Using the Tagsort Option:

Sorting is one of the more computationally intensive operations. It requires a lot of hard drive space, which can be a problem, especially for data sets with many observations and a large number of variables. You can be more efficient in sorting if you use the tagsort option. This method sorts only the key variables and then rebuilds the dataset by pulling up the appropriate observation and attaching all the rest of the variables. The tagsort method often takes longer to sort a data set, but uses less hard drive space.  

proc sort data=march_recode tagsort;

     by date dest;

run;

Creating a New Sorted Data Set:

If you wish to create a new data set, and maintain the input data set in its original order, you can use the out= option on the Proc Sort statement, as shown below:

proc sort data=march_recode out=sortdat;

     by totpassngrs;

run;

Getting Rid of Duplicate Cases for the Same ID – Using the Nodupkey Option:

Proc Sort provides an easy way to get rid of duplicate cases having the same values of the key variables.  Use the nodupkey option on the Proc Sort statement, as shown below.  Check the log to see how many duplicates were deleted. The original data set will not be affected.

proc sort data=march_recode out=sortdat2 nodupkey;

   by date dest;

run;

Getting Rid of Duplicate Records for the Same ID – Using the Noduprec Option:

You can also ask SAS to eliminate any cases that are duplicates for all variables using the noduprec option, as shown in the code below.  

proc sort data=march_recode out=sortdat3 noduprec; 

   by date dest; 

run;
Selecting cases for analysis:
Cases can be selected for a given analysis by using a where statement. 

Selecting cases based on values of a character variable:

When selecting cases based on the value of a character variable, be sure to enclose the value or values in quotes, as shown below:

title "Flights to Los Angeles";

proc print data=march_recode;

    where dest = "LAX";

    var flight dest totpassngrs;

run;

If you wish to select the observations based on a missing value for a character variable, use quotes around a blank " " (the missing value for character variables).

title "Missing Destination";

proc print data=march_recode;

    where dest = " ";

    var flight dest totpassngrs; run;

Selecting cases based on values of a numeric variable:

Cases used in an analysis may be selected based on the values of a numeric variable. The Boolean operators (<, >, <=, >=, =, ~=) may be used to get the desired case selection, as shown below. Do not use quotes when specifying the value of a numeric variable.

title "Flights Less than 30 Percent Full";

proc print data=march_recode;

    where pctfull < 30;

    var dest date pctfull;run;

Those with PCTFULL missing are also included in this case selection, because missing is evaluated as less than any numeric value.   

                              Flights Less than 30 Percent Full                             

                              Obs    dest          date    pctfull

                               99    ORD     03/05/1990    28.0952

                              102    WAS     03/05/1990    17.2222

                              235    DFW     03/12/1990    27.2222

                              390    WAS     03/19/1990    29.4444

                              421    LAX     03/21/1990      .

                              451    WAS     03/22/1990    18.3333

                              512    WAS     03/25/1990    23.8889

The syntax below can be used to exclude the missing values from those cases printed:

title "Flights Less than 30 Percent Full";

proc print data=march_recode;

    where pctfull not=. and pctfull < 30;

    var dest date pctfull;

run;

The where statement can also be used with “between” to restrict cases used in an analysis. The example below will print those cases with percent full from 25 to 35:

title "Flights Between 25 and 35 Percent Full";

proc print data=march_recode;

    where pctfull between 25 and 35;run;

If you wish to select observations based on a missing value for a numeric variable, use a period to indicate missing, as shown in the example below.  

title "Cases Where Number of Passengers is Missing";

proc print data=march_recode;

    where totpassngrs = .;

    var flight dest totpassngrs;run;

You can also select cases using a combination of character and numeric variables in the where statement:

title "Flights less than 60 percent full to London";
proc print data=march_recode;

    where (pctfull < 60) and (dest="LON") ;

    var flight dest totpassngrs capacity pctfull;run;

Selecting cases based on dates:

You can select cases for a procedure based on dates, by using a SAS date constant.  Note that the date constant is specified in quotes with the day as a two-digit number, followed by a three-letter abbreviation for the month, followed by a 2 or 4-digit number for the year. A letter D (either upper or lower case) must appear after the quote to let SAS know that this is a date.

title "Flights on March 7th, 1990"; 

proc print data=march_recode;

    where date = "07MAR90"D;run;

You can also use "where … between" with dates to specify a range of dates:

title "Flights March 7th to March 9th , 1990"; 

proc print data=march_recode;

    where date between "07MAR90"D and "14MAR90"D; 

run;

You can use the same method for selecting observations based on missing values for a date variable as for a numeric variable, because dates are stored as numeric values in SAS.

title "Cases with Missing Date";

proc print data=march_recode;

    where date = .;


    var flight dest date;

run;

Basic One-Sample and Two-Sample Statistical Tests Using SAS
(commands=basic_stat.sas)
The examples in this handout use the PULSE data, which is available in the labdata folder as an Excel file (PULSE.XLS), as a raw data file (PULSE.DAT), and as a CSV file (PULSE.CSV). Information on the variables in the PULSE data is included in the data set descriptions.
Read in the raw data for PULSE:
data pulse;

   infile "pulse.csv" firstobs=2 delimiter="," missover;

   input pulse1 pulse2 ran smokes sex height weight activity;

   label pulse1 = "Resting pulse, rate per minute"


   pulse2 = "Second pulse, rate per minute";

run;

/*Print out the first 25 cases*/

proc print data=pulse (obs=25);

run;

Create and assign formats to variables:
proc format;


value sexfmt 1="Male" 2="Female";


value yesnofmt  1="Yes" 2="No";


value actfmt 1="Low" 2="Medium" 3="High";

run;

proc print data=pulse (obs=25) label;

format sex sexfmt. ran smokes yesnofmt. activity actfmt.;

run;

Output from these commands is shown below:
          Resting         Second

        pulse, rate    pulse, rate

 Obs     per minute     per minute    ran    smokes     sex      height    weight    activity

   1         64             88        Yes     No       Male        66        140      Medium

   2         58             70        Yes     No       Male        72        145      Medium

   3         62             76        Yes     Yes      Male        73        160      High

   4         66             78        Yes     Yes      Male        73        190      Low

   5         64             80        Yes     No       Male        69        155      Medium

   6         74             84        Yes     No       Male        73        165      Low

   7         84             84        Yes     No       Male        72        150      High

   8         68             72        Yes     No       Male        74        190      Medium

   9         62             75        Yes     No       Male        72        195      Medium
/*Display data file information and variable lists */
proc contents data=pulse varnum;

run;

                                     The CONTENTS Procedure

         Data Set Name        WORK.PULSE                       Observations          92

         Member Type          DATA                             Variables             8

         Engine               V9                               Indexes               0

         Created              Tue, Sep 12, 2006 02:54:34 PM    Observation Length    64

         Last Modified        Tue, Sep 12, 2006 02:54:34 PM    Deleted Observations  0

         Protection                                            Compressed            NO

         Data Set Type                                         Sorted                NO

         Label

         Data Representation  WINDOWS_32

         Encoding             wlatin1  Western (Windows)

                                Engine/Host Dependent Information

Data Set Page Size          8192

Number of Data Set Pages    1

First Data Page             1

Max Obs per Page            127

Obs in First Data Page      92

Number of Data Set Repairs  0

File Name                   C:\DOCUME~1\kwelch\LOCALS~1\Temp\SAS Temporary

                            Files\_TD1744\pulse.sas7bdat

Release Created             9.0101M3

Host Created                XP_PRO

                                   Variables in Creation Order

                 #    Variable    Type    Len    Label

                 1    pulse1      Num       8    Resting pulse, rate per minute

                 2    pulse2      Num       8    Second pulse, rate per minute

                 3    ran         Num       8

                 4    smokes      Num       8

                 5    sex         Num       8

                 6    height      Num       8

                 7    weight      Num       8

                 8    activity    Num       8
proc means data=pulse; 

run;

                                         The MEANS Procedure

 Variable  Label                            N          Mean       Std Dev       Minimum       Maximum

 ----------------------------------------------------------------------------------------------------

 pulse1    Resting pulse, rate per minute  92    72.8695652    11.0087052    48.0000000   100.0000000

 pulse2    Second pulse, rate per minute   92    80.0000000    17.0937943    50.0000000   140.0000000

 ran                                       92     1.6195652     0.4881540     1.0000000     2.0000000

 smokes                                    92     1.6956522     0.4626519     1.0000000     2.0000000

 sex                                       92     1.3804348     0.4881540     1.0000000     2.0000000

 height                                    92    68.7391304     3.6520943    61.0000000    75.0000000

 weight                                    92   145.1521739    23.7393978    95.0000000   215.0000000

 activity                                  92     2.1195652     0.5711448     1.0000000     3.0000000

 ----------------------------------------------------------------------------------------------------
One-Sample Tests:

Binomial Confidence Intervals and Tests for Binary Variables:

If you have a categorical variable with only two levels, you can use the binomial option to request a 95% confidence interval for the proportion in the first level of the variable. In the PULSE data set, SMOKES=1 indicates those who were smokers, and SMOKES=2 indicates non-smokers. Use the (p=) option to specify the null hypothesis proportion that you wish to test for the first level of the variable. In the commands below, we test hypotheses for the proportion of SMOKES=1 (i.e., proportion of smokers) in the population. By default SAS produces an asymptotic test of the null hypothesis:

H0: proportion of smokers = 0.25
HA: proportion of smokers ( 0.25
proc freq data = pulse;

   tables  smokes / binomial(p=.25);

run;
                                                            smokes
                                                      Cumulative    Cumulative

                   smokes    Frequency     Percent     Frequency      Percent

                   -----------------------------------------------------------

                        1          28       30.43            28        30.43

                        2          64       69.57            92       100.00

                                     Binomial Proportion

                                         for smokes = 1

                                --------------------------------

                                Proportion                0.3043

                                ASE                       0.0480

                                95% Lower Conf Limit      0.2103

                                95% Upper Conf Limit      0.3984

                                Exact Conf Limits

                                95% Lower Conf Limit      0.2127

                                95% Upper Conf Limit      0.4090

                                 Test of H0: Proportion = 0.25

                                ASE under H0              0.0451

                                Z                         1.2039

                                One-sided Pr >  Z         0.1143

                                Two-sided Pr > |Z|        0.2286

                                        Sample Size = 92

If you wish to obtain an exact binomial test of the null hypothesis, use the exact statement. 

proc freq data = pulse;

   tables  smokes / binomial(p=.25);

   exact binomial;

run;

This results in an exact test of the null hypothesis, in addition to the default asymptotic test.

                                      Exact Test

                               One-sided Pr >=  P           0.1399

                               Two-sided = 2 * One-sided    0.2797
Chi-square Goodness of Fit Tests for Categorical Variables:

Use the chisq option in the tables statement to get a chi-square goodness of fit test, which can be used for categorical variables with two or more levels. By default SAS assumes that you wish to test the null hypothesis that the proportion of cases is equal in all categories. In the variable ACTIVITY, a value of 1 indicates a low level of activity, a value of 2 is a medium level of activity, and a value of 3 indicates a high level of activity.

proc freq data = pulse;

   tables activity / chisq; 

run;

                                          activity

                                                       Cumulative    Cumulative

                  activity    Frequency     Percent     Frequency      Percent

                  -------------------------------------------------------------

                         1          10       10.87            10        10.87

                         2          61       66.30            71        77.17

                         3          21       22.83            92       100.00

                                         Chi-Square Test

                                      for Equal Proportions

                                      ---------------------

                                      Chi-Square    46.9783

                                      DF                  2

                                      Pr > ChiSq     <.0001

                                        Sample Size = 92
If you wish to specify your own proportions, use the testp = option in the tables statement.  This option allows you to specify any proportions that you wish to test for each level of the variable in the tables statement, as long as the sum of the proportions equals 1.0. In the example below we test the null hypothesis:

H0: P1 = 0.20, P2=.50, P3=.30
proc freq data = pulse;

   tables activity  /chisq testp = ( .20 , .50, .30 );

run;

                                      The FREQ Procedure

                                             activity

                                                    Test     Cumulative    Cumulative

            activity    Frequency     Percent     Percent     Frequency      Percent

            -------------------------------------------------------------------------

                   1          10       10.87       20.00            10        10.87

                   2          61       66.30       50.00            71        77.17

                   3          21       22.83       30.00            92       100.00

                                         Chi-Square Test

                                    for Specified Proportions

                                    -------------------------

                                    Chi-Square        10.3043

                                    DF                      2

                                    Pr > ChiSq         0.0058

                                        Sample Size = 92

You may also specify percentages to test, as long as they add up to 100 percent:

proc freq data = pulse;

   tables activity  /chisq testp = ( 20 , 50, 30 );

run;
One-Sample test for a continuous variable:
You can use Proc Univariate to carry out a one-sample t-test to test the population mean against any null hypothesis value you specify by using mu0= option. The default, if no value of mu0 is specified is that mu0 = 0. In the commands below, we test:

H0: (0=72

HA: (0(72

Note that SAS also provides the non-parametric Sign test and Wilcoxon signed rank test.

proc univariate data=pulse mu0=72;


var pulse1;


histogram / normal (mu=est sigma=est);


qqplot /normal (mu=est sigma=est);

run;
Selected output from Proc Univariate:

                                      Proc Univariate

                                  Tests for Location: Mu0=72

                           Test           -Statistic-    -----p Value------

                           Student's t    t  0.757635    Pr > |t|    0.4506

                           Sign           M        -3    Pr >= |M|   0.5900

                           Signed Rank    S      96.5    Pr >= |S|   0.6797

Equivalently, we can carry out a one-sample t-test in Proc Ttest by specifying the H0= option.:

proc ttest data=pulse H0=72 ;

  var pulse1;

run;
                      Variable:  pulse1  (Resting pulse, rate per minute

                  N        Mean     Std Dev     Std Err     Minimum     Maximum

                 92     72.8696     11.0087      1.1477     48.0000       100.0

                    Mean       95% CL Mean        Std Dev      95% CL Std Dev

                  72.8696     70.5897  75.1494     11.0087      9.6155  12.8779

                                     DF    t Value    Pr > |t|

                                     91       0.76      0.4506
Chi-Square Tests of Independence:

Two by Two Table:

If you wish to examine the relationship between two categorical variables, you can use Proc Freq. Use the chisq option to obtain the Pearson chi-square test of independence (or of homogeneity), and use the expected option to get the expected value in each cell. The commands below can be used to get a cross-tabulation. In this case, we have a 2 by 2 table, because each categorical variable has two levels. We test:

H0: SEX is independent of SMOKING STATUS
HA: SEX is not independent of SMOKING STATUS

proc freq data=pulse;

  tables sex*smokes / chisq;

run;

Note that Fisher’s exact test is produced by default for a 2 x 2 table, when the chisq option is specified. Read either the one-sided or two-sided p-value for the Fisher’s exact test:

                                    The FREQ Procedure

                                     Table of sex by smokes

                               sex       smokes

                               Frequency|

                               Expected |

                               Percent  |

                               Row Pct  |

                               Col Pct  |       1|       2|  Total

                               ---------+--------+--------+

                                      1 |     20 |     37 |     57

                                        | 17.348 | 39.652 |

                                        |  21.74 |  40.22 |  61.96

                                        |  35.09 |  64.91 |

                                        |  71.43 |  57.81 |

                               ---------+--------+--------+

                                      2 |      8 |     27 |     35

                                        | 10.652 | 24.348 |

                                        |   8.70 |  29.35 |  38.04

                                        |  22.86 |  77.14 |

                                        |  28.57 |  42.19 |

                               ---------+--------+--------+

                               Total          28       64       92

                                           30.43    69.57   100.00

                              Statistics for Table of sex by smokes

                     Statistic                     DF       Value      Prob

                     ------------------------------------------------------

                     Chi-Square                     1      1.5321    0.2158

                     Likelihood Ratio Chi-Square    1      1.5699    0.2102

                     Continuity Adj. Chi-Square     1      1.0089    0.3152

                     Mantel-Haenszel Chi-Square     1      1.5154    0.2183

                     Phi Coefficient                       0.1290

                     Contingency Coefficient               0.1280

                     Cramer's V                            0.1290

                                      Fisher's Exact Test

                               ----------------------------------

                               Cell (1,1) Frequency (F)        20

                               Left-sided Pr <= F          0.9310

                               Right-sided Pr >= F         0.1576

                               Table Probability (P)       0.0886

                               Two-sided Pr <= P           0.2502

                                        Sample Size = 92
R by C Table:

If you have a table with more than two rows, and/or more than two columns, you can still specify chi-square test of independence by using the chisq option. In this example, we have a 2 by 3 table.

proc freq data = pulse;

   tables sex * activity / chisq expected nocol nopercent;

run;

                                The FREQ Procedure

                                    Table of sex by activity

                          sex       activity

                          Frequency|

                          Expected |

                          Row Pct  |       1|       2|       3|  Total

                          ---------+--------+--------+--------+

                                 1 |      6 |     35 |     16 |     57

                                   | 6.1957 | 37.793 | 13.011 |

                                   |  10.53 |  61.40 |  28.07 |

                          ---------+--------+--------+--------+

                                 2 |      4 |     26 |      5 |     35

                                   | 3.8043 | 23.207 | 7.9891 |

                                   |  11.43 |  74.29 |  14.29 |

                          ---------+--------+--------+--------+

                          Total          10       61       21       92

                             Statistics for Table of sex by activity

                     Statistic                     DF       Value      Prob

                     ------------------------------------------------------

                     Chi-Square                     2      2.3641    0.3067

                     Likelihood Ratio Chi-Square    2      2.4827    0.2890

                     Mantel-Haenszel Chi-Square     1      1.4339    0.2311

                     Phi Coefficient                       0.1603

                     Contingency Coefficient               0.1583

                     Cramer's V                            0.1603

/*SAME TEST, BUT for non-smokers only*/

proc freq data=pulse;


tables sex * activity /nopercent nocol expected chisq;


where smokes=2;run;

                                          The FREQ Procedure

                                       Table of sex by activity

                             sex       activity

                             Frequency|

                             Expected |

                             Row Pct  |Low     |Medium  |High    |  Total

                             ---------+--------+--------+--------+

                             Male     |      3 |     22 |     12 |     37

                                      | 3.4688 | 24.281 |   9.25 |

                                      |   8.11 |  59.46 |  32.43 |

                             ---------+--------+--------+--------+

                             Female   |      3 |     20 |      4 |     27

                                      | 2.5313 | 17.719 |   6.75 |

                                      |  11.11 |  74.07 |  14.81 |

                             ---------+--------+--------+--------+

                             Total           6       42       16       64

                               Statistics for Table of sex by activity

                        Statistic                     DF       Value      Prob

                        ------------------------------------------------------

                        Chi-Square                     2      2.5961    0.2731

                         WARNING: 33% of the cells have expected counts less

                                  than 5. Chi-Square may not be a valid test.
Note the warning message in the output above, indicating that the chi-square test may not be appropriate, due to small expected counts in 33% of the cells of the table. In this case, you can use a Fisher’s exact test, by specifying the exact statement, as shown below:
/*For tables containing sparse cells, Fisher's exact test can be performed */

proc freq data=pulse;


tables sex * activity /nopercent nocol;


exact fisher;


where smokes=2;

run;
Note: read the p-value for Fisher’s exact test from the output labeled Pr <=P, at the bottom of the output, as shown in the selected output below.

                                       Fisher's Exact Test

                                  ----------------------------------

                                  Table Probability (P)       0.0221

                                  Pr <= P                     0.2568

If your variables have a large number of categories, or if your data set is large, you may want to use a Monte Carlo simulation to get the p-value for the Fisher’s exact test, rather than take the time and resources to calculate the exact p-value. To do this, use the mc option in the exact statement. Specify the random seed (which needs to be a positive integer <=32767) using the seed= option. Note: read the p-value from the line that says Pr <= P, not the Table Probability, which is actually the hypergeometric probability for the sample data table only.
proc freq data = pulse;
   where smokes=2;
   tables sex * activity / chisq expected;

   exact fisher / mc seed=1234;

run;
                            Statistics for Table of sex by activity

                                      Fisher's Exact Test

                               ----------------------------------

                               Table Probability (P)       0.0221

                             Monte Carlo Estimate for the Exact Test

                               Pr <= P                     0.2594

                               99% Lower Conf Limit        0.2481

                               99% Upper Conf Limit        0.2707

                               Number of Samples            10000

                               Initial Seed                  1234

                                        Sample Size = 64

Cross-Tabulations for Two Categorical Variables, Stratified by Another Variable:

To get a cross-tabulation of three variables (i.e., two variables stratified by a third variable) use syntax similar to that shown below. This syntax will produce two separate tables of SEX by ACTIVITY, one for each level of RAN. This type of syntax can be extended to produce higher-way cross-tabulations. The table produced by SAS will always be formed by the last two variables listed. All prior variables will be used to form the strata. 

proc freq data = pulse;

   tables ran*sex*activity / chisq;

run;

Two-Sample Tests for a Continuous Variable:

Independent Samples t-test:

If you wish to compare the means of a continuous variable for two independent groups, you can carry out an independent samples t test. Some of the assumptions for this test are:
1) The two groups are independent.

2) The variable whose means we are interested in should be roughly normally distributed within each group. You can do a Q-Q plot to check normality for each group.

3) The variability of the observations in each of the groups should be similar. A side-by-side boxplot is helpful to check the equal variance assumption. 

/*Q-Q plot can be used to check normality assumption of test variables, at each level of the grouping variable */

proc sort data=pulse;


by ran;

run;

proc univariate data=pulse;


by ran;


var pulse2;


histogram / normal;


qqplot /normal (mu=est sigma=est);

run;

/*Boxplot allows you to compare the test variables VISUALLY between two groups, for both location and spread */
proc sort data=pulse;

  by ran;

proc boxplot data=pulse;


plot (pulse1 pulse2) * ran;

run;

/*Independent samples t test*/

proc ttest data=pulse;


var pulse1 pulse2;


class ran;

run;
                                 The TTEST Procedure

                                            Statistics

                                Lower CL          Upper CL  Lower CL           Upper CL

   Variable  ran             N      Mean    Mean      Mean   Std Dev  Std Dev   Std Dev  Std Err

   pulse1    Yes            35    69.672    73.6    77.528    9.2501   11.436    14.983    1.933

   pulse1    No             57    69.551  72.421    75.291    9.1319   10.817    13.269   1.4327

   pulse1    Diff (1-2)           -3.537  1.1789    5.8951    9.6488   11.055    12.944   2.3739

   pulse2    Yes            35    86.007  92.514    99.022    15.323   18.943    24.819    3.202

   pulse2    No             57    69.676  72.316    74.955     8.399   9.9484    12.204   1.3177

   pulse2    Diff (1-2)           14.208  20.198    26.189    12.255   14.041     16.44   3.0152

                                               T-Tests

                Variable    Method           Variances      DF    t Value    Pr > |t|

                pulse1      Pooled           Equal          90       0.50      0.6207

                pulse1      Satterthwaite    Unequal        69       0.49      0.6257

                pulse2      Pooled           Equal          90       6.70      <.0001

                pulse2      Satterthwaite    Unequal      45.7       5.83      <.0001

                                        Equality of Variances

                    Variable    Method      Num DF    Den DF    F Value    Pr > F

                    pulse1      Folded F        34        56       1.12    0.6990

                    pulse2      Folded F        34        56       3.63    <.0001
Wilcoxon rank sum test:

If you are unwilling to assume normality for your test variable or the sample size is too small for you to appeal to the central-limit-theorem, you may want to use non-parametric tests. The Wilcoxon rank sum test (also known as the Mann-Whitney test) is the non-parametric analog of the independent sample t test.

/*Non-parametric test: Wilcoxon/Mann-whitney test*/

proc npar1way data=pulse wilcoxon;


class ran;


var pulse1 pulse2;

run;
Output for PULSE2 is shown below:

                                        The NPAR1WAY Procedure

                           Wilcoxon Scores (Rank Sums) for Variable pulse2

                                     Classified by Variable ran

                                    Sum of      Expected       Std Dev          Mean

                 ran       N        Scores      Under H0      Under H0         Score

                 -------------------------------------------------------------------

                 Yes      35        2292.0       1627.50    124.056919     65.485714

                 No       57        1986.0       2650.50    124.056919     34.842105

                                 Average scores were used for ties.

                                  Wilcoxon Two-Sample Test

                                   Statistic             2292.0000

                                   Normal Approximation

                                   Z                        5.3524

                                   One-Sided Pr >  Z        <.0001

                                   Two-Sided Pr > |Z|       <.0001

                                   t Approximation

                                   One-Sided Pr >  Z        <.0001

                                   Two-Sided Pr > |Z|       <.0001

                              Z includes a continuity correction of 0.5.
A Monte-Carlo approximation of the exact p-value can be obtained for the Wilcoxon test by using an exact statement, as shown below:

proc npar1way data=pulse wilcoxon;


class ran;


var pulse1 pulse2;

      exact wilcoxon / mc;

run;
This results I the following additional output for PULSE2. Note that if no seed is given, a seed based on the clock is used.

                         Monte Carlo Estimates for the Exact Test

                             One-Sided Pr >= S

                             Estimate                        0.0000

                             99% Lower Conf Limit            0.0000

                             99% Upper Conf Limit         4.604E-04

                             Two-Sided Pr >= |S - Mean|

                             Estimate                        0.0000

                             99% Lower Conf Limit            0.0000

                             99% Upper Conf Limit         4.604E-04

                             Number of Samples                10000

                             Initial Seed                 297742738
Paired Samples t-test:

If you wish to compare the means of two variables that are paired (i.e. correlated), you can use a paired sample t-test for continuous variables. You may first wish to test to see if the variables are in fact correlated.

/*PAIRED SAMPLES TESTS*/

proc corr data=pulse;

  var pulse1 pulse2;

run;

                                   The CORR Procedure

                                2  Variables:    pulse1   pulse2

                                       Simple Statistics

   Variable           N          Mean       Std Dev           Sum       Minimum       Maximum

   pulse1            92      72.86957      11.00871          6704      48.00000     100.00000

   pulse2            92      80.00000      17.09379          7360      50.00000     140.00000

                                       Simple Statistics

                           Variable    Label

                           pulse1      Resting pulse, rate per minute

                           pulse2      Second pulse, rate per minute

                           Pearson Correlation Coefficients, N = 92

                                   Prob > |r| under H0: Rho=0

                                                         pulse1        pulse2

                   pulse1                               1.00000       0.61620

                   Resting pulse, rate per minute                      <.0001

                   pulse2                               0.61620       1.00000

                   Second pulse, rate per minute         <.0001

Proc ttest can be used with a paired statement, to get a paired samples t-test:

proc ttest data=pulse;

  paired pulse2*pulse1;

run;

                                     The TTEST Procedure

                                           Statistics

                           Lower CL          Upper CL  Lower CL           Upper CL

   Difference           N      Mean    Mean      Mean   Std Dev  Std Dev   Std Dev  Std Err

   pulse2 - pulse1     92    4.3406  7.1304    9.9203    11.766   13.471    15.759   1.4045

                                            T-Tests

                         Difference           DF    t Value    Pr > |t|

                         pulse2 - pulse1      91       5.08      <.0001

The paired t-test can be carried out for each level of RAN. The commands and results of these commands are shown below:

proc sort data=pulse;

  by ran;

run;

proc ttest data=pulse;

  paired pulse2*pulse1;

  by ran;

run;
   ---------------------------------------- ran=1 ---------------------------------------------

                                       The TTEST Procedure

                                           Statistics

                           Lower CL          Upper CL  Lower CL           Upper CL

   Difference           N      Mean    Mean      Mean   Std Dev  Std Dev   Std Dev  Std Err

   pulse2 - pulse1     35    13.745  18.914    24.084    12.173    15.05    19.718   2.5439

                                           T-Tests

                         Difference           DF    t Value    Pr > |t|

                         pulse2 - pulse1      34       7.44      <.0001

--------------------------------------------- ran=2 ---------------------------------------------

                                       The TTEST Procedure

                                           Statistics

                           Lower CL          Upper CL  Lower CL           Upper CL

   Difference           N      Mean    Mean      Mean   Std Dev  Std Dev   Std Dev  Std Err

   pulse2 - pulse1     57    -1.209  -0.105    0.9987    3.5126   4.1605    5.1039   0.5511

                                            T-Tests

                         Difference           DF    t Value    Pr > |t|

                         pulse2 - pulse1      56      -0.19      0.8492

McNemar’s test for paired categorical data:

If you wish to compare the proportions in a 2 by 2 table for paired data, you can use McNemar’s test, by specifying the agree option in Proc Freq. Before running the McNemar’s test, we recode PULSE1 and PULSE2 into two categorical variables HIPULSE1 and HIPULSE2, as shown below:

data newpulse;

  set pulse;

  if pulse1 > 80 then hipulse1 = 1;

  if pulse1 > 0 and pulse1 <=89 then hipulse1=0;

  if pulse2 > 80 then hipulse2 = 1;

  if pulse2 > 0 and pulse2 <=89 then hipulse2=0;

run;

proc freq data=newpulse;

  tables hipulse1 hipulse2;

run;
                                       The FREQ Procedure

                                                       Cumulative    Cumulative

                  hipulse1    Frequency     Percent     Frequency      Percent

                  -------------------------------------------------------------

                         0          82       89.13            82        89.13

                         1          10       10.87            92       100.00

                                                       Cumulative    Cumulative

                  hipulse2    Frequency     Percent     Frequency      Percent

                  -------------------------------------------------------------

                         0          71       77.17            71        77.17

                         1          21       22.83            92       100.00

We can now carry out McNemar’s test of symmetry to see if the proportion of participants with a high value of PULSE1 is different than the proportion of participants with a high value of PULSE2.

proc freq data=newpulse;

  tables hipulse1*hipulse2/ agree;

run;
                                  Table of hipulse1 by hipulse2

                               hipulse1     hipulse2

                               Frequency|

                               Percent  |

                               Row Pct  |

                               Col Pct  |       0|       1|  Total

                               ---------+--------+--------+

                                      0 |     69 |     13 |     82

                                        |  75.00 |  14.13 |  89.13

                                        |  84.15 |  15.85 |

                                        |  97.18 |  61.90 |

                               ---------+--------+--------+

                                      1 |      2 |      8 |     10

                                        |   2.17 |   8.70 |  10.87

                                        |  20.00 |  80.00 |

                                        |   2.82 |  38.10 |

                               ---------+--------+--------+

                               Total          71       21       92

                                           77.17    22.83   100.00

                          Statistics for Table of hipulse1 by hipulse2

                                         McNemar's Test

                                     -----------------------

                                     Statistic (S)    8.0667

                                     DF                    1

                                     Pr > S           0.0045

                                        Sample Size = 92

Statistical Graphics 
Using SAS 9.2

This handout describes the use of SAS procedures to create basic statistical graphs. The procedures introduced are:

· Proc Sgplot

· Proc Sgpanel

· Proc Sgscatter

These procedures are new to SAS 9.2.  Check the SAS online documentation for more information. The files produced are .png (portable network graphics) files, which can be easily imported into other applications. Examples using traditional SAS Graphics are also included for comparison and to illustrate different methods of creating/saving the files.

Using the Employee Data Set

The permanent SAS dataset employee.sas7bdat, is used for these examples, and can be downloaded from the web site:

http://www.umich.edu/~kwelch/ 
· Save this data set in a folder on your desktop (or any other location you choose). Do not double-click to open it. 

· Submit a libname statement to point to the folder (not the actual file) where you have saved the data set. The libname statement only needs to be submitted once when you start SAS. Be sure to change the path to the directory you want to use.

libname mylib "c:\" ;
· Check the log to see that the library (libref) was successfully assigned. 

     libname mylib "c:\" ;

NOTE: Libref WORKSHOP was successfully assigned as follows:

      Engine:        V9

      Physical Name: c:\

Statistical Graphics Examples
Boxplots

title "Boxplot";

title2 "No Categories";

proc sgplot data=mylib.employee;

  vbox salary;

run; 

title "Boxplot";

title2 "Category=Gender";

proc sgplot data=mylib.employee;

  vbox salary/ category=gender;

run; 

Paneled Boxplots

title "Boxplot with Panels";

proc sgpanel data=mylib.employee;

  panelby jobcat / rows=1 columns=3 ;

  vbox salary / category= gender;

run;

Barcharts

title "Vertical Bar Chart";

proc sgplot data=mylib.employee;

  vbar jobcat ;

run;

Stacked Bar Charts

title "Vertical Bar Chart";

title2 "Grouped by Gender";

proc sgplot data=mylib.employee;

  vbar jobcat /group=Gender;

run;

Bar Chart with Mean and Error Bars

title "BarChart with Mean and Standard Deviation";

proc sgplot data=mylib.employee;

  vbar jobcat / response=salary limitstat = stddev

  limits = upper stat=mean;

run;

Bar Charts for Proportions of a Binary Variable
/*Bar chart with Mean of Indicator Variable*/

data afifi;

  set mylib.afifi;

  if survive=1 then died=0;

  if survive=3 then died=1;

run;

proc format;

  value shokfmt 2="Non-Shock"

                3="Hypovolemic"  

                4="Cardiogenic"

                5="Bacterial"

                6="Neurogenic"           

                7="Other";        

run;

title "Barchart of Proportion Died for each Shock Type";

proc sgplot data=afifi;

  vbar shoktype / response=died stat=mean;

  format shoktype shokfmt.;

run;

Paneled Bar Charts
title "BarChart Paneled by Gender";

proc sgpanel data=mylib.employee;

   panelby gender ;

   vbar jobcat / response=salary limitstat = stddev

      limits = upper stat=mean;

run;

Histograms

title "Histogram";

proc sgplot data=mylib.employee;

  histogram salary ;

run; 

Histogram with Density Overlaid

title "Histogram With Density Overlaid";

proc sgplot data=mylib.employee;

  histogram salary ;

  density salary;

  density salary / type=kernel;
  keylegend / location = inside position = topright; run;

Paneled Histograms

title "Histogram with Panels";

title2 "Exclude Custodial";

proc sgpanel data=mylib.employee;

  where jobcat not=2;

  panelby   gender jobcat/  rows=2 columns = 2;

  histogram salary / scale=proportion; run;
/*use scale=proportion, count, or percent(default)*/

Overlaid Histograms

title "Overlay different variables";

proc sgplot data=mylib.employee;

  histogram salbegin ;

  histogram salary / transparency = .5;

run;

/*Create New Variables for Overlay*/
data employee2;

  set mylib.employee;

  if gender = "m" then salary_m = salary;

  if gender = "f" then salary_f = salary;

run;

title "Overlaid histograms";

title2 "Same variable, but two groups ";

proc sgplot data=employee2;

  histogram salary_m;

  histogram salary_f / transparency=0;

run;
Note: Transparency = 0 is opaque. Transparency = 1.0 is fully transparent.

title "Overlaid histograms";

title2 "Logarithmic Scale for X";

proc sgplot data=employee2;

  histogram salary_m / scale=proportion ;

  histogram salary_f / scale=proportion transparency=0;

  Xaxis type=log logbase=10 logstyle=linear;run;

Scatterplots

title "Scatterplot";

proc sgplot data=mylib.employee;

  scatter x=salbegin y=salary / group=gender ;

run;

Scatterplot with Confidence Ellipse

title "Scatterplot";

proc sgplot data=mylib.employee;

  scatter x=salbegin y=salary / group=gender ;
  ellipse x=salbegin y=salary / type=predicted alpha=.10;
run;

Scatterplot with Regression Line

title "Scatterplot with Regression Line";
title2 "Clerical Only";
proc sgplot data=mylib.employee;
  where jobcat=1;
  scatter x=prevexp y=salary / group=gender ;
  reg x=prevexp y=salary / cli clm;
run;

Scatterplot with Separate Regression Lines for Subgroups

title "Scatterplot with Regression Line";
title2 "Separate Lines for Females and Males";
proc sgplot data=mylib.employee;
  where jobcat=1;
  reg x=prevexp y=salary / group=gender;
run;

Paneled Scatterplots with Loess Fit

title "Scatterplot Panels";
title2 "Loess Fit";
proc sgpanel data=mylib.employee;

   panelby jobcat;

   scatter x=jobtime y=salary / group=gender;

   loess x=jobtime y=salary ;

run;

Scatterplot Matrix
title "Scatterplot Matrix"; 

title2 "Clerical Employees";

proc sgscatter data=mylib.employee;

  where jobcat=1;

  matrix salbegin salary jobtime prevexp / group=gender
                   diagonal=(histogram kernel);

run;

Series plots

The next plot uses the autism dataset (Oti, Anderson, and Lord, 2007). We first import the .csv file using Proc Import.

/*Series Plots*/

PROC IMPORT OUT= WORK.autism 

            DATAFILE= "autism.csv" 

            DBMS=CSV REPLACE;

     GETNAMES=YES;

     DATAROW=2; 

RUN;

title "Spaghetti Plots for Each Child";

proc sgpanel data=autism;

  panelby sicdegp /columns=3;

  series x=age y=vsae / group=Childid 

  markers  legendlabel=" " lineattrs=(pattern=1 color=black); run;

Overlay Means on Plots
We can calculate the means by SICDEGP and AGE and overlay these means on a dot plot of the raw data using the commands below.

proc sort data=autism;

by sicdegp age;

run;

proc means data=autism noprint;

  by sicdegp age;

  output out=meandat mean(VSAE)=mean_VSAE;

run;

data autism2;

  merge autism meandat(drop=_type_ _freq_);

  by sicdegp age;

run;

title "Means Plots Overlaid on Data";

proc sgplot data=autism2;

  series x=age y=mean_VSAE / group=SICDEGP;

  scatter x=age y=VSAE ;

run;

Using formats to make graphs more readable

This example creates two formats, the first one is numeric (i.e. it can be used to format a numeric variable), and the second one is character (i.e., it can be used to format a character variable). These are temporary formats and must be submitted each time you run SAS. Use a format statement to apply the formats for each proc that you run, where appropriate.

proc format;

   value jobcat 1="Clerical"

                2="Custodial"




3="Manager";

   value $Gender "f"="Female"

                 "m"="Male";

run;

title "Boxplot with Panels";

proc sgpanel data=mylib.employee;

  panelby jobcat / rows=1 columns=3 novarname;

  vbox salary / category= gender ;

  format gender $gender.;

  format jobcat jobcat.;

run;

Editing ODS Graphs

You can edit graphs by going to the command dialog box and typing sgedit on. This will allow you to right-click on a given graph, and then select EDIT. From the edit window that is opened, you can modify the Style of the graph, insert text, insert rows, etc.  You can only edit graphs that were submitted after the sgedit on command.

Creating pdf output

You may wish to create pdf output. This can be accomplished by setting up the ODS environment, so the output goes directly into a .pdf file. The Journal2 style is a grayscale type of output. Other styles are available. If you use: ods listing close; SAS will not produce a .png file. If you omit that statement, both a .png and a .pdf file will be created. To be sure the .png output is created again after the graph is completed, use: ods listing.

ods pdf style=journal2;

ods pdf file = "testing.pdf";

ods listing close;

title "PDF Output";

proc sgpanel data=mylib.employee;

   panelby jobcat;

   scatter x=jobtime y=salary / group=gender;

   loess x=jobtime y=salary ;

run;

ods pdf close;

ods listing;
Saving Statistical Graphics plots from Sgplot, Sgscatter, and Sgpanel

Graphs generated using Statistical Graphics procedures (Proc Sgplot, Proc Sgscatter, Proc Sgpanel) will show up in your SAS Current Folder in Windows, and will automatically be .png files. These graphics files are very compact, because they use raster graphics, and can be easily imported into windows applications, such as Microsoft Word, or PowerPoint.

You do not need to export graphs created using Statistical Graphics procedures. They will automatically be saved to your Current Folder in Windows as .png files. You can double-click on the .png files to view them, or you can view them as thumbnails. They will be given names such as SGPlot.png, or SGPlot1.png, etc. 

To set the Current Folder, double-click on the location listed at the bottom of your SAS desktop and browse to the folder you wish to use. Make sure you have double-clicked on the name of the folder. Do this before you submit the SAS commands to create the Statistical Graphs.

Within SAS, graphs created using Statistical Graphics procedures will not go to your SAS/Graph window, but instead will be in the Results window. Double-click on the procedure name and then double-click on the individual graphs. You can browse forward and backward through the graphs once you have created them. 

[image: image33.png]~: SAS - [save_sasgraphs_9_2.sas]
[ Fie Edt Vew Took Run Solutions Window Help Bl

2

vV DEH S8R B0 DA £ X0&

Results.

& res

= (@9 oo Distrbution o Sl
3] The SGPlo Procedure

= (@9 Saplok Regresson lt
0] The 5GPl Procedure.

libname mylib "c:\documents and settings\kwelch\des

title "Distribution of salary”:

=proc sgplot data=mylib.basebal!
histogram salary:

run;

=data baseball2;
set mylib.baseballs
log_salary = log(salary):
run;

title "Regression Plot with Log salary”:
=proc sgplot data=baseballZ:

reg x=no_hits y=log_salary / cli clm:
run; quit;

|
Output - (Untitled] | ] Log - {untitled) | [# save_sasgraphs_9...

autosave complete

3 C:iDocuments and SettingsikwelchiDesktopimylb n17, Col1





[image: image34]
[image: image35.png]Fle Edt View Favortes Took Help

Qs - © - ] Do iy ros |-

adress | C:\Dosuments and SettingsikwelchiDesktoplworkshop

File and Folder Tasks (%

) Rename tis e
5 Move tis e
) copy tisfie

@ Fubishthis e o the ~ employeesas7hdat
web

) Emaiths e

& Printthis fie
X Delete tisile

Other Places
SGPlt13.ng SGPanelt.prg

Details

5GPanel3.prg

testing pof

SGPlt12png





Traditional Graphics Examples 

The following instructions show how to create traditional graphics in the SAS/Graph window using Proc Univariate and Proc Gplot. These graphs can be produced using either SAS 9.1 or 9.2.

Creating a Histogram Using Proc Univariate

title "Distribution of Salary";

proc univariate data=mylib.employee noprint;

  var salary;

  histogram;

run;

Creating a Regression Plot Using Proc Gplot

symbol1 value=dot height=.5 interpol=rl ;

title "Regression Plot for Salary";

proc gplot data=mylib.employee;

  plot salary * prevexp ;

run; quit;

Saving Traditional graphs from the Graph Window

Graphs generated using Proc Gplot or Proc Univariate will appear in the SAS/Graph window. You can Export these graphs to a file format that can be read by any windows applications that can read graphics files. You can save SAS graphs from the graphics window using any of the commonly used formats for graphs supported by SAS (.bmp, .gif, .tif).  You can also save graphics files from the SAS/Graph window using a .png (portable network graphics) format. 

Go to the SAS/Graph window. With the appropriate graph open in the Graph Window, Go to File...Export as Image....Select the  File type you want (e.g. .png), Browse to the location where you wish to save the graphics file, and type the file name, e.g.
  histogram_salary.png 

Bringing graphics files into a Word document
You can simply drag and drop a graphics file into word, or you can import it using the steps shown below:

Make sure you are not at the beginning or end of a document, or it will be difficult to work with the graph. Place your mouse somewhere in the middle of several blank lines in the document. Go to Insert…Picture from file… Browse until you get to your graph (e.g., histogram_salary.png).  

You can resize the graph by clicking your mouse anywhere in the graph to get the outline. Then grab the lower right corner with your mouse (you should see an arrow going northwest to southeast) and move it up and to the left to make it smaller, or down and to the right to make it larger. You can't easily edit the graph in Word. If you're using a .png file, you can simply drag and drop it into Word.

Bringing graphics files into PowerPoint

You can drag and drop a graphics file into PowerPoint, or you can import it using the steps shown below:

Insert a new slide. Go to File...Insert...Picture. Browse until you get to your graph,

histogram_salary.png

Click on Insert.

Note: any titles you used in SAS will appear on your graph.

You can resize the graph by clicking your mouse anywhere in the graph to get the outline. Then grab the lower right corner with your mouse (you should see an arrow going northwest to southeast) and move it up and to the left to make it smaller, or down and to the right to make it larger. You can't easily edit the graph in Power Point.

Have fun!!

Double-click here in the Results window to view the graph.





Set the current folder by double-clicking here. Graphs  created by Statistical Graphics procedures will automatically be saved here.

